

Please remove this section from the newsletter (pages VIII.1 through VIII.22) and use it to replace the original "Weather" section of the DOE-2.1A Reference Manual

DOE-2 Weather Processor

Fred Buhl

LBL Simulation Research Group

April 1999

Weather Data in DOE-2.1E	2
Introduction	2
Weather Variables	2
Hourly Variables	2
Hourly Solar Variables	2
Monthly Variables	2
Header Data	2
Source Data and Formats	2
Source Data	2
Formats: TMY2, WYEC2, CD144 and TRY	3
Solar Data	3
Weather Processor	4
Input and Output	4
Input Description	4
ASCII Weather Files	7
Moving Data	7
INPUT.TMP for FMTWTH	7
FMTWTH Program Listing	7
WTHFMT Program Listing	9
Processing Nonstandard Weather Data	10
Use of the subroutine OTHER	10
Example of subroutine OTHER	11
FORTRAN fragment: direct and diffuse solar calculation	13
Using Standard Formats for Measured Data	15
Using the DOE-2 Measured Weather Format	15
TRY Format	17
INDEX	21

Weather Data in DOE-2.1E

Introduction

The Loads and HVAC simulations in DOE-2 require hourly weather data, which are contained in DOE-2 weather files. These weather files are created from source hourly data by the DOE-2 weather processor program *doewth*. We describe here the weather variables used by DOE-2, the types and formats of source data, the various ways to use *doewth*, and how to process measured weather data or data in nonstandard formats.

Weather Variables

The weather variables used by DOE-2 and present on the weather file are as follows.

Hourly Variables

The DOE-2 weather file contains hourly data for one year (8760 hours). Leap years are ignored; all DOE-2 weather files are 365 days long. The hourly data on the weather files are:

Dry-bulb Temperature (F)
Wet-bulb Temperature (F)
Atmospheric Pressure (inches of Hg times 100)
Wind Speed (knots)
Wind Direction (compass points 0-15, with 0 being north, 1 NNE, etc.)
Cloud Amount (0 to 10, with 0=clear and 10=totally overcast)
Cloud Type (0, 1, or 2) 0 is cirrus or cirrostratus, the least opaque;
 1 is stratus or stratus fractus, the most opaque; and
 2 is all other cloud types, of medium opacity
Humidity Ratio (lb of water per lb of dry air)
Density of the Air (lb/ft³)
Specific Enthalpy of the Air (Btu/lb)
Rain Flag (0 means it is not raining; 1 means it is)
Snow Flag (0 means it is not snowing; 1 means it is)

Hourly Solar Variables

There are two types of DOE-2 weather files: those with hourly solar values and those without. In the case of the files without solar data, DOE-2 calculates solar values using the ASHRAE clear sky model and the clearness numbers, cloud amounts, and cloud types from the DOE-2 weather file. The solar DOE-2 weather files contain the following hourly values:

- Total Horizontal Solar Radiation (Btu/hr-ft²)
- Direct Normal Solar Radiation (Btu/hr-ft²)

Monthly Variables

- Clearness Number (dimensionless). This is the ASHRAE clearness number; see, for example, p. 27.12, Figure 7, of the 1993 ASHRAE Handbook of Fundamentals.
- Ground Temperature (Rankine)

Header Data

The header record on the weather file contains the file identification, latitude, longitude, and time zone.

Source Data and Formats

Source Data

Source weather data for building energy simulation programs can be broken into two major classes: *historical data* and *typical weather years*. Historical data is just "real" data: usually measured (but sometimes modeled) data from a particular location for a given period of record. Typical years are ersatz years assembled to match the long term data from a particular location using a particular statistical measure.

The primary source for historical weather data is the U.S. National Climatic Data Center (NCDC)¹. NCDC can provide hourly historical data for thousands of locations around the world. This data may not always be complete; data items or periods of record may be missing. A highly reliable source of historical data for U.S. locations is the Solar and Meteorological Surface Observational Network (SAMSON) data set assembled by the National Renewable Energy Laboratory (NREL)². This contains a 30 year (1961 to 1990) period of record for 239 locations.

There are several sets of typical year data that are widely used in DOE-2 simulations. The TMY2 data set was derived from the SAMSON database by NREL. There are typical years for 239 locations in the US and its territories. NREL also developed the WYEC2 typical year data set for ASHRAE. The selection criteria were different for TMY2 and WYEC2, with TMY2 weighting the solar data more heavily. The California Energy Commission (CEC) has defined 16 climate zones with corresponding typical year data sets (CTZ2) for use in Title 24 energy code compliance. National Resources Canada (NRC) has funded WATSUN Energy Laboratory at the University of Waterloo to create typical weather years for Canada. There are files currently available for 51 locations.

Formats: TMY2, WYEC2, CD144 and TRY

Source data comes in various formats. Typically the files are ASCII, but the data items, units, item location, and record length vary from format to format. NCDC can provide historical data in a variety of formats: TD-3280, TD-9950 and TD-1440 (CD144). Of these, the DOE-2 weather processor can only process CD144, usually called TD-1440 by NCDC. Typical years now come in two formats: **TMY2** and **WYEC2**. Note that TMY2 and WYEC2 are names of both typical year data sets and data formats. The DOE-2 weather processor can process files in either TMY2 or WYEC2 format. The NREL TMY2 typical weather year data sets are in TMY2 format. The ASHRAE WYEC2, the Canadian CWEC, and the CEC CTZ2 typical weather year data sets are all in WYEC2 format. One other format worth mentioning is **TRY**. This is the format of an old, typical year data set that did not include solar radiation data. The format can still be useful, however, as a format for measured data. The DOE-2 weather processor can process data in this format and add modeled solar data to the DOE-2 weather file. The format is described below, under "Processing Nonstandard Weather Data."

Solar Data

Source weather data files may or may not contain solar data. All TMY2 and WYEC2 files contain solar data. The weather processor will transfer this data to the DOE-2 weather file and it will be used by the DOE-2 simulation program. Historical weather data files in CD144 format do not contain solar data nor is such data generally available for a specific location and time period. In this case, ersatz solar data must be generated from cloud cover and other data using sky models and regression formulas. This solar data generation can be done either in the weather processor or in the DOE-2 program itself. *When the DOE-2 program detects a DOE-2 weather file containing no solar data it will generate the solar data needed for the simulation.* However, it is more efficient and accurate to have the weather processor generate the solar data. To accomplish this, put CD144S instead of CD144 in line 3 of the weather processor input. For TRY format source data, put TRYSLM instead of TRY. CD144S or TRYSLM tells the weather processor to generate solar data and put it on the DOE-2 weather file for use by the DOE-2 simulation program.

TMY2

The following description is quoted from NREL's website at http://rredc.nrel.gov/solar/old_data/nsrdb/tmy2/:

The TMY2s are data sets of hourly values of solar radiation and meteorological elements for a one-year period. Their intended use is for computer simulations of solar energy conversion systems and building systems to facilitate performance comparisons of different system types, configurations, and locations in the United States and its territories. Because they represent typical rather than extreme conditions, they are not suited for designing systems to meet the worst-case conditions occurring at a location.

To distinguish between the old and new TMY data sets, the new TMY data sets are referred to as TMY2s. TMY and TMY2 data sets cannot be used interchangeably because of differences in time (solar versus local), formats, elements, and units. Unless they are revised, computer programs designed for TMY data will not work with TMY2 data.

The TMY2 data sets and manual were produced by the National Renewable Energy Laboratory's (NREL's) Analytic Studies Division under the Resource Assessment Program, which is funded and monitored by the U.S. Department of Energy's Office of Solar Energy Conversion.

¹ National Climatic Data Center, Federal Building, 151 Patton Avenue, Asheville, NC 28801-5001, <http://www.ncdc.noaa.gov>.

² National Renewable Energy Laboratory, 1617 Cole Boulevard, Golden, CO ;80401-3393, <http://www.nrel.gov>.

WYEC2

The following is quoted from the Watsun Simulation Lab website at <http://dial.uwaterloo.ca/~watsun/cwed.htm>:

Canadian weather for energy calculations (CWEC) files are typical year sets of meteorological data in WYEC2 format; they include such quantities as solar radiation (global, diffuse, direct), dry-bulb, and dew point temperatures, wind speed and direction, atmospheric pressure, etc., on an hourly basis. They were developed by the Watsun Simulation Laboratory under the auspices of the National Research Council of Canada.

The following description is quoted from the Watsun Simulation Lab website at <http://dial.waterloo.ca/~watsun/cwecovvw.htm>:

The CWEC files are created by concatenating twelve Typical Meteorological Months selected from a database of, in most cases, 30 years of data. The method is similar to TMY procedure developed in the 1980s by Sandia National Laboratory. The months are chosen by statistically comparing individual with long-term monthly means for daily total global radiation, mean-, minimum- and maximum- dry-bulb temperatures, mean-, minimum- and maximum- dew point temperatures, and mean and maximum wind speeds. The composite index used to select the most 'typical' months uses the following weights (%):

Parameter	Dry Bulb Max	Dry Bulb Min	Dry Bulb Mean	Dew Point Max	Dew Point Min	Dew Point Mean	Wind Speed Max	Wind Speed Mean	Daily Solar Rad.
Weight(%)	5	5	30	2.5	2.5	5	5	5	40

Additional consideration is given, in the selection process, to the statistics and persistence structures of the daily mean dry-bulb temperature and daily total radiation. A complete description of the procedure used can be found in: D.L. Siurna, L.J. D'Andrea, K.G.T. Hollands, "A Canadian Representative Meteorological Year for Solar System Simulation," Proceedings of the 10th Annual Conference of the Solar Energy Society of Canada (SESCI '84), August 1-6, 1984, Calgary, Alberta, Canada.

In the CWEC files, no missing values will be found in the following WYEC2 fields: extraterrestrial irradiance (101), global horizontal irradiance (102), direct normal irradiance (103), diffuse horizontal irradiance (104), weather (204), station pressure (205), dry-bulb temperature (206), dew point temperature (207), wind direction (208), wind speed (209), total sky cover (210), opaque sky cover (211), snow cover (212). The original long-term data sets (up to 40 years of data) from which the CWEC files were derived can also be obtained directly from Environment Canada.

Weather Processor

The DOE-2 weather processor is a batch or command-line program called *doewth* or *doewth.exe*, depending on the computing environment. The primary function of the weather processor is to read hourly weather data in a variety of formats, extract the data needed by DOE-2, and write a packed binary weather file that is used by the DOE-2 simulation program. In addition to its primary function (called packing) the weather processor can produce hourly listings of raw or packed weather files in a readable format and can produce a summary report of the data on a packed DOE-2 weather file.

Input and Output

The weather processor requires two input files. One, called WEATHR.TMP, is the hourly weather data. This will be either an ASCII (text) file, if raw weather data is being packed, or a binary file, if a packed DOE-2 file is being listed or summarized. The second input file, INPUT.TMP, is a short ASCII file that tells the weather processor what functions it is to perform and supplies any additional information needed to perform the task. This file is described in detail below. Output is on two files. The file OUTPUT. contains any listings, reports, and error messages. The file NEWTH.TMP contains the packed DOE-2 weather file, if one is being created. Copying, renaming, and saving these files will normally be performed by a procedure file specific to the computing environment being used.

Input Description

Here we describe the contents of the input file (INPUT.TMP) needed to perform the three functions:

- create a DOE-2 weather file (packing) (*PACK*),
- listing a weather file (*LIST*) and
- producing a statistical summary (*STAT*).

PACK

- line 1: The word PACK in columns 1-4.
- line 2: The station name in columns 1-20. This name will be written on the output file as identification. The entry here is for the user only and is arbitrary.
- line 3: The data is entered as shown below. When the format is shown as L, it signifies that the datum must be left justified in the columns indicated. The format R signifies that the datum must be right justified in the columns indicated, and the format D means that the value should be entered with a decimal point (neither right or left justification is required). For those with FORTRAN background: L corresponds to A6, R to I6, and D to F6.1.

Example of how the data is entered (line 3)

Columns	Format	Description
1-6	L	A code-word specifying the unpacked file type. Options are TMY2, WYEC2, CD144, CD144S ^a , TRY, TRYSLM ^a , TD9685, and OTHER ^b .
7-12	R	Weather station number. This is required.
Note: for TMY2 files, the following inputs on line 3 may be left blank		
13-18	R	The year of the weather data (e.g., 1999). This is required for CD144 and TD9685 files (which can contain several years of weather data). For other files, -999 should be input.
19-24	R	Time zone (as in the SITE-PARAMETERS command)
25-30	D	Latitude (degrees). Positive north of the equator, negative south of the equator.
31-36	D	Longitude (degrees). Positive west of Greenwich, negative east of Greenwich.
37-42	L	A code-word specifying the number of bits per word to be used in packing the output file. The options are 60-BIT or 30-BIT (for 32-bit machines)
43-48	L	A code-word specifying the type of output file. The options are NORMAL and SOLAR. NORMAL produces a DOE-2 weather file with no solar data. SOLAR produces a file containing solar information.
49-54	R	Interpolation interval. The program fills in missing data by linear interpolation between the last and the next value present, if the number of hours of missing data is less than or equal to the interpolation interval. If more hours of data are missing than the interpolation interval, it still does interpolation up to 24 hours and a warning message is issued. If more than 24 hours are missing, the previous value is used. The interpolation interval must be less than 24 ^c .
55-60	D	This sets the maximum dry-bulb temperature change allowed in one hour. Changes larger than this will cause a warning message to be printed.
61-66	D	Soil thermal diffusivity (ft ² /hr). Used for calculating monthly ground temperatures. A value of 0.010 can be used for dry soil, 0.025 for average soil, and 0.050 for wet soil.
67-72	D	Station altitude (feet), used in CD144 and TD9685.
73-78	R	Location needed only for CD144S and TRYSLM to choose a cloud cover model. See ILOC. Used only for CD144 and TRY formats. Select the location that best represents the data being packaged.
^a CD144S tells the weather processor to read a file in CD144 format and add ersatz solar data using the ASHRAE clear sky model, SOLMET cloud cover regressions formula, and the Erbs-Klein-Duffie direct/diffuse model. TRYSLM does the same for data in TRY formats.		^b If OTHER is chosen, the data should either be in the DOE-2 measured weather data format (see Processing Nonstandard Weather Data) or a special OTHER processing subroutine must be written and installed in the weather processor. To accomplish the latter, the you must have the source code and a FORTRAN compiler.
		^c The weather processor makes no evaluation of the data to see that it is internally consistent, except that during interpolation it never allows the wet-bulb temperature to exceed the dry-bulb temperature, or the dew point temperature to exceed the wet-bulb temperature.

ILOC and Station Name			
01 ALBUQUERQUE, NM	08 CHARLESTON, SC	15 GREAT FALLS, MT	21 NEW YORK, NY
02 APALACHICOLA, FL	09 COLUMBIA, MO	16LAKE CHARLES, LA	22 NORTH OMAHA, NE
03 BISMARCK, ND	10 DODGE CITY, KS	17 MADISON, WI	23 PHOENIX, AZ
04 BOSTON, MA	11 EL PASO, TX	18 MEDFORD, OR	24 SANTA MARIA, CA
05 BROWNSVILLE, TX	12 ELY, NV	19 MIAMI, FL	25 SEATTLE-TACOMA,
06 CAPE HATTERAS, NC	13 FORT WORTH, TX	20 NASHVILLE, TN	26 WASHINGTON, DC
07 CARIBOU, ME	14 FRESNO, CA		

- line 4: Contains the 12 clearness numbers (one per month) in D format in column intervals 1-6, 7-12, 13-18, etc. (skip for TMY2; unused for WYEC2, so can be just 1.0). See 1993 ASHRAE Fundamentals, p. 27.12.
- line 5: Contains the 12 ground temperatures (one per month in F) in D format in column intervals 1-6, 7-12, 13-18, etc. (skip for TMY2). A value of -999 will flag the program to calculate the ground temperature using the method of Kusuda and Achenbach (ASHRAE Trans. 41 (1965) p. 61).

LIST

line 1: The word LIST in columns 1-4.

line 2:

Columns	Format	Description
1-6	L	Input type or file type. Options are PACKED, OTHER, TRY, WYEC2, CD144, TMY2 and TD9685.
7-12	R	Year of weather data (e.g., 1972). Required for CD144 and TD9685 files. Should be -999 if columns 1-6 is TRY, TMY2, WYEC2 or PACKED.
13-18	R	Station number. PACKED can have -999 entered here.
19-24	R	Beginning month of listing (1 to 12 for January to December)
25-30	R	Ending month of listing (1 to 12 for January to December)

STAT

Only one card is necessary for the STAT option with the word STAT in columns 1-4. STAT produces a statistical summary (average monthly temperatures, etc.) of the data on the weather file.

Multiple Options

To exercise several options in the same computer run simply concatenate the input for several functions. The last card in the entire file, whether running one or several options, must have the word END in columns 1-3.

Examples

1. To PACK, LIST for 12 months, and STAT a Mexico City CD144 file.

```

1234567890123456789012345678901234567890123456789012345678901234567890
line 1:  PACK
line 2:  MEXICO CITY 90
line 3:  CD144S 76679 1990      6 19.24 99.0930-BITSOLAR      4 20 .025 7329. 20
line 4:  0.9 0.9 0.9 0.9 0.9 0.9 0.9 0.9 0.9 0.9 0.9 0.9
line 5:  -999.
line 6:  LIST
line 7:  PACKED -999 -999      1 12
line 8:  STAT
line 9:  END

```

2. To LIST the month of January 1990 from a Mexico City CD144 file.

```

123456789012345678901234567890123456789012345678901234567890123456
line 1:  LIST
line 2:  CD144 1990 76679      1 1
line 3:  END

```

3. TMY2 input.

```
PACK
```

```
Atlanta GA TMY2
TMY2 13874
STAT
END
```

ASCII Weather Files

Moving Data

This topic involves the transfer of packed weather files from one computer system to another. The weather files used by DOE-2 are packed, binary files and, therefore, they cannot be transferred from computer to computer unless the computer type and operating systems are identical. Frequently, the original raw data used to produce the packed files has been lost or discarded. When a new computer system is installed, or when it is necessary to do runs at another site, a method is needed to preserve and transfer the data contained in the packed files. To do this, we run a small program called WTHFMT, which reads a packed binary weather file and writes out an ASCII file containing the same information. The ASCII file can then be written to floppy disk, along with a program called FMTWTH, which reverses the process (i.e., read the ASCII weather file and output a packed binary DOE-2 compatible file suitable for the new computer system). The disk recipient just has to read the disk, compile FMTWTH and execute it with the ASCII weather file as input. Both WTHFMT and FMTWTH are in FORTRAN and will compile and execute on a Sun workstation and on an IBM-compatible PC. For other systems, the code may need minor changes (to the OPEN statements, for instance).

The ASCII weather file can be examined and edited on a terminal, using the local editor. This means that the technique can be used when you want to make changes to a packed weather file. The format of the ASCII file has been chosen to be easily readable by humans.

With reference to the programs that follow, FMTWTH reads a one-line input file (INPUT.TMP) in addition to the file containing the weather data. INPUT.TMP tells FMTWTH what type of packed binary file to produce. Note that only the first line is used; the subsequent lines are explanatory. The numbers on the first line must be in columns 13 and 31, respectively.

INPUT.TMP for FMTWTH

```
-----+-----1-----+-----2-----+-----3-----+-----4-----+-----5-----+-----6-----+-----7
WORD SIZE = 2      FILE TYPE = 2
WORD SIZE = 1 MEANS 60-BIT, 2 MEANS 30-BIT
FILE TYPE = 1 MEANS OLD, 2 MEANS NORMAL (NO SOLAR DATA),
3 MEANS THE DATA HAS SOLAR DATA
```

FMTWTH Program Listing

```
PROGRAM FMTWTH
C
C   THIS PROGRAM READS A FORMATTED WEATHER FILE (WEATHER.FMT)
C   AND A FORMATTED INPUT FILE (FMTWTH.INP) AND WRITES A
C   PACKED BINARY DOE2 WEATHER FILE (WEATHER.BIN)
C
C   DIMENSION CLN(12),GT(12),MDAYS(12),IDAT(1536),IWDID(5)
C
C   DATA MDAYS / 31,28,31,30,31,30,31,31,30,31,30,31 /
C
C   OPEN(UNIT=12,FILE='INPUT.DAT')
C   OPEN(UNIT=11,FILE='WEATHER.FMT')
C   OPEN(UNIT=10,FILE='WEATHER.BIN',FORM='UNFORMATTED')
C
C   IWSZ          WORD SIZE          1 = 60-BIT, 2 = 30-BIT
C   IFTYP         FILE TYPE          1 = OLD, 2 = NORMAL (NO SOLAR),
C                                     3 = THE DATA HAS SOLAR
C   IWDID         LOCATION I.D.
C   IWYR          YEAR
C   WLAT          LATITUDE
C   WLONG         LONGITUDE
C   IWTZN         TIME ZONE NUMBER
C   IWSOL         SOLAR FLAG         FUNCTION OF IWSZ + IFTYP
C   CLN           CLEARNESS NO.
```

```

C      GT          GROUND TEMP.      (DEG R)
C      KMON        MONTH              (1-12)
C      KDAY        DAY OF MONTH
C      KH          HOUR OF DAY
C      WBT         WET-BULB TEMP      (DEG F)
C      DBT         DRY-BULB TEMP      (DEG F)
C      PATM        PRESSURE           (INCHES OF HG)
C      CLDAMT      CLOUD AMOUNT       (0 - 10)
C      ISNOW       SNOW FLAG          (1 = SNOWFALL)
C      IRAIN       RAIN FLAG          (1 = RAINFALL)
C      IWNDDR      WIND DIRECTION     (0 - 15; 0=N, 1=NNE, ETC)
C      HUMRAT      HUMIDITY RATIO     (LB H2O/LB AIR)
C      DENSTY      DENSITY OF AIR     (LB/CU FT)
C      ENTHAL      SPECIFIC ENTHALPY  (BTU/LB)
C      SOLRAD      TOTAL HOR. SOLAR    (BTU/HR-AREA)
C      DIRSOL      DIR. NORMAL SOLAR  (BTU/HR-AREA)
C      ICLDTY      CLOUD TYPE         (0 - 2)
C      WNDSPD      WIND SPEED         KNOTS
C
      REWIND 12
      READ (12,9001) IWSZ,IFTYP
9001  FORMAT(12X,I1,17X,I1)
      REWIND 11
      READ (11,9002) (IWDID(I),I=1,5),IWYR,WLAT,WLONG,IWTZN,IWSOL
9002  FORMAT(5A4,I5,2F8.2,2I5)
      IWSOL = IWSZ + (IFTYP-1)*2 - 1
      READ (11,9003) (CLN(I),I=1,12)
      READ (11,9004) (GT(I),I=1,12)
9003  FORMAT(12F6.2)
9004  FORMAT(12F6.1)
      DO 1000 IM=1,12
      IDE = MDAYS(IM)
      DO 1000 ID=1,IDE
      IRECXO = IM*2 + (ID-1)/16 - 1
      IDXO = MOD(ID-1,16) + 1
      DO 500 IH=1,24
      READ (11,9005) KMON, KDAY, KH, WBT, DBT, PATM, CLDAMT, ISNOW,
1      IRAIN, IWNDDR, HUMRAT, DENSTY, ENTHAL, SOLRAD,
2      DIRSOL, ICLDTY, WNDSPD
9005  FORMAT(3I2,2F5.0,F6.1,F5.0,2I3,I4,F7.4,F6.3,F6.1,2F7.1,I3,F5.0)
      ISOL = INT(SOLRAD + .5)
      IDN = INT(DIRSOL + .5)
      IWET = INT(WBT+99.5)
      IDRY = INT(DBT+99.5)
      IPRES = INT(PATM*10.-149.5)
      ICLDAM = INT(CLDAMT)
      IWNDSP = INT(WNDSPD+0.5)
      IHUMRT = INT(HUMRAT*10000.+0.5)
      IDENS = INT(DENSTY*1000.-19.5)
      IENTH = INT(ENTHAL*2.0+60.5)
      IP1 = (IDXO-1)*96 + IH*4 - 3
      IDAT(IP1) = IPRES*65536 + IWET*256 + IDRY
      IDAT(IP1+1) = ISOL*1048576 + IDN*1024 +
1      ICLDAM*64 + ISNOW*32 + IRAIN*16 + IWNDDR
      IDAT(IP1+2) = IHUMRT*128 + IDENS
      IDAT(IP1+3) = IENTH*2048 + ICLDTY*128 + IWNDSP
500  CONTINUE
      IF ((ID .NE. 16) .AND. (ID .NE. IDE)) GO TO 1000
      WRITE (10) IWDID,IWYR,WLAT,WLONG,IWTZN,IRECXO,IDE,CLN(IM),
1      GT(IM),IWSOL,IDAT
1000 CONTINUE
      END
-----1-----2-----3-----4-----5-----6-----7

```

WTHFMT Program Listing

```

PROGRAM WTHFMT
C
C   THIS PROGRAM READS A PACKED BINARY DOE-2 WEATHER FILE AND
C   CREATES A FORMATTED WEATHER FILE AS OUTPUT.  THE INPUT
C   FILE IS WEATHER.BIN, THE OUTPUT FILE IS WEATHER.FMT.
C
      DIMENSION CLN(12),GT(12),MDAYS(12),IDAT30(1536),
      IWDID(5),IWTH(14)
      DIMENSION XMASK(16,2), CALC(16)
      INTEGER IDUM
C
      DATA MDAYS / 31,28,31,30,31,30,31,31,30,31,30,31 /
      DATA IWDID /5*4H /
      DATA XMASK / -99., -99., 15., 0., 0., 0., 0., 0., .02, -30., 0.,
1      .0, .0, .0, .0, 10.,
2      1., 1., .1, 1., 1., 1., 1., .0001, .001, .5,
3      1., 1., 1., 1., 0., 0. /
C
      OPEN (UNIT=11,FILE='WEATHER.FMT')
      OPEN (UNIT=10,FILE='WEATHER.BIN',FORM='UNFORMATTED')
      REWIND 10
      DO 100 IM1=1,12
      READ (10) (IWDID(I),I=1,5),IWYR,WLAT,WLONG,IWTZN,LRECX,NUMDAY,
      CLN(IM1),GT(IM1),IWSOL
      READ (10) IDUM
100 CONTINUE
      REWIND 10
      LRECX = 0
      WRITE (11,9001) (IWDID(I),I=1,5),IWYR,WLAT,WLONG,IWTZN,IWSOL
      WRITE (11,9002) (CLN(I),I=1,12)
      WRITE (11,9003) (GT(I),I=1,12)
9001 FORMAT(5A4,I5,2F8.2,2I5)
9002 FORMAT(12F6.2)
9003 FORMAT(12F6.1)
      DO 1000 IM2=1,12
      IDE = MDAYS(IM2)
      DO 1000 ID=1,IDE
      DO 1000 IH=1,24
105  IRECX = IM2*2 + (ID-1)/16 - 1
      IDX = MOD(ID-1,16) + 1
      IF (IRECX-LRECX) 200,400,300
200  IDIF = LRECX - IRECX + 1
      DO 220 I=1,IDIF
      BACKSPACE 10
220 CONTINUE
300 READ (10) IWDID,IWYR,WLAT,WLONG,IWTZN,LRECX,NUMDAY,CLRNES,
      TGRND,IDUM,IDAT30
      GO TO 105
400 CONTINUE
      IP1 = 96*(IDX-1) + 4*IH - 3
      IWTH(3) = IDAT30(IP1)/65536
      IWTH(1) = MOD(IDAT30(IP1),65536)/256
      IWTH(2) = MOD(IDAT30(IP1),256)
      IWTH(11) = IDAT30(IP1+1)/1048576
      IWTH(12) = MOD(IDAT30(IP1+1),1048576)/1024
      IWTH(4) = MOD(IDAT30(IP1+1),1024)/64
      IWTH(5) = MOD(IDAT30(IP1+1),64)/32
      IWTH(6) = MOD(IDAT30(IP1+1),32)/16
      IWTH(7) = MOD(IDAT30(IP1+1),16)
      IWTH(8) = IDAT30(IP1+2)/128
      IWTH(9) = MOD(IDAT30(IP1+2),128)
      IWTH(10) = IDAT30(IP1+3)/2048
      IWTH(13) = MOD(IDAT30(IP1+3),2048)/128

```

```

      IWTH(14) = MOD(IDAT30(IP1+3),128)
      DO 500 I=1,14
      CALC(I) = FLOAT(IWTH(I))*XMASK(I,2) + XMASK(I,1)
500  CONTINUE
      ISNOW = INT(CALC(5) + .01)
      IRAIN = INT(CALC(6) + .01)
      IWNDDR = INT(CALC(7) + .01)
      ICLDTY = INT(CALC(13) + .01)

C
C          IM2          MOMTH          (1-12)
C          ID          DAY OF MONTH
C          IH          HOUR OF DAY
C          CALC(1)     WET-BULB TEMP    (DEG F)
C          CALC(2)     DRY-BULB TEMP    (DEG F)
C          CALC(3)     PRESSURE          (INCHES OF HG)
C          CALC(4)     CLOUD AMOUNT      (0 - 10)
C          ISNOW       SNOW FLAG         (1 = SNOWFALL)
C          IRAIN       RAIN FLAG         (1 = RAINFALL)
C          IWNDDR      WIND DIRECTION     (0 - 15; 0=N, 1=NNE, ETC)
C          CALC(8)     HUMIDITY RATIO    (LB H2O/LB AIR)
C          CALC(9)     DENSITY OF AIR    (LB/CU FT)
C          CALC(10)    SPECIFIC ENTHALPY (BTU/LB)
C          CALC(11)    TOTAL HOR. SOLAR  (BTU/HR-AREA)
C          CALC(12)    DIR. NORMAL SOLAR (BTU/HR-AREA)
C          ICLDTY     CLOUD TYPE         (0 - 2)
C          CALC(14)    WIND SPEED        KNOTS
C
      900 WRITE (11,9005) IM2, ID, IH, CALC(1), CALC(2), CALC(3), CALC(4),
      1          ISNOW, IRAIN,IWNDDR, CALC(8), CALC(9), CALC(10),
      2          CALC(11), CALC(12), ICLDTY, CALC(14)
9005 FORMAT(3I2,2F5.0,F6.1,F5.0,2I3,I4,F7.4,F6.3,F6.1,2F7.1,I3,F5.0)
1000 CONTINUE
      ENDFILE 11
      END

```

Processing Nonstandard Weather Data

The DOE-2 weather processor is capable of processing raw weather data in a variety of formats into a DOE-2 compatible form. Quite frequently, however, users obtain weather data in a format that is unknown to the weather processor. The user then has two alternatives: convert the data into a known format; or process the raw weather data directly by writing code for subroutine OTHER in the DOE-2 weather processor.

Use of the subroutine OTHER

OTHER is a typical weather data processing subroutine in the DOE-2 weather processor. Like the other such routines (TRYDCD, TMYDCD, etc.) it is called once every 24 hours by subroutine PACKER, and its use can be triggered by the weather processor input. Putting OTHER in the first 5 columns of the third line of the PACK input sequence informs the weather packer that subroutine OTHER will be used for reading in and processing the raw data. OTHER is meant to be a user programmable subroutine for processing special weather formats. OTHER already contains code which processes data in the DOE-2 measured weather format. The user can use this format, or replace the code in OTHER with code that will read a different format using the original OTHER code as an example. Basically, the arrays in the common block /RAWDAT/ must be filled for each call to OTHER. The arrays are dimensioned 24, and are all integers. They are

IDRY	dry-bulb temperature in °F, rounded to the nearest whole degree.
IWET	wet-bulb temperature in °F, rounded to the nearest whole degree.
IDEW	dew point temperature in °F, rounded to the nearest whole degree.
IPRESS	atmospheric pressure in inches of Hg times 100 (so 29.92 will be 2992, etc.)
IWNDS	wind speed, in knots (!) (nearest whole knot).
ICLAMT	cloud amount (sky cover), 0 - 10; 0 = no clouds, 10 = totally overcast.
ISOL,	total solar on a horizontal surface and direct normal (beam) solar radiation, both in Btu/ft ² -hr,
IDN	nearest whole unit.
IWNDR	wind direction in compass points (0 - 15). 0 is north, 1 is NNE, 15 is NNW.

ICLTY	DOE-2 cloud type; takes the values 0, 1, and 2. 0 = most transparent cloud category. It corresponds to TRY types 8 and 9 (cirrus and cirrostratus or cirrocumulus). 1 = most opaque. It corresponds to TRY type 2 (stratus or fractus stratus). 2 = intermediate transparency. It corresponds to all other types of clouds, and is a good default if no cloud type information is available.
ICLTY1	DOE-2 cloud type - UNUSED
IRN, ISN	the rain and snow flags, respectively. Set to 1 if raining or snowing, 0 otherwise. IRN and ISN are not currently used in DOE-2, but it is nice to set them anyway. When printed out along with the other weather variables with the LIST option of the weather processor, they can help explain some otherwise odd looking weather or solar data.

Frequently the raw data will include dry-bulb and relative humidity, instead of dry-bulb, wet-bulb, and dew point as required by the weather processor. You should use the procedure given in 1993 ASHRAE Fundamentals, p. 6.14, situation number 3. A slightly different procedure is used in the example OTHER subroutine, below.

In the case of solar data, the weather processor needs total horizontal and direct normal. The data is often in the form of direct and diffuse on a horizontal surface. The cosine of the solar zenith angle (or the sine of the solar altitude) must then be calculated in order to obtain the direct normal from the direct horizontal. This calculation involves knowing the solar declination angle and the equation of time; it is complicated by the fact that the cosine of the zenith angle must usually be averaged over a one hour time bin, since the solar data point is usually the average over one hour of a number of data points taken at less than one-hour intervals. In this case it is best to simply follow the procedure shown in the following example subroutine. As in the example, it is usually necessary to do a limit check on the resulting direct normal, particularly at sunrise and sunset, where the data is frequently bad.

Sometimes only total horizontal solar data is available. A model must then be employed to obtain the direct radiation from the total. We recommend the model of Erbs, Klein, and Duffie (*Solar Energy*, 28, (1982) p. 293. See “*FORTTRAN fragment: direct and diffuse solar calculation*,” below, or the code included in the weather processor for processing the DOE-2 measured weather data format.

Example of subroutine OTHER

```

SUBROUTINE OTHER
C
C          INSERT YOUR ROUTINE TO DECODE SPECIAL TAPE FORMAT
C
C          COMMON /LOCALD/  STALAT, STALON, SSTALA, CSTALA, TSTALA, HRSLON
C
C          COMMON /CONST/   DTOR, PIOVR2, PIOVR4, PIOV12, IBLNK
C
C          COMMON /TIMES/   IMNTH, IDAY, I HOUR, IRECXO, IDXO, IDAYL, ITIM
C
C          COMMON /MONTHC/  BEFORE(12), MDAYS(12), MNAMES(12)
C          INTEGER          BEFORE
C
C          COMMON /FILES/   INFIL, OUTFIL, INWTH, OUTWTH, SOLWTH, STOUT
C          INTEGER          OUTFIL, OUTWTH, SOLWTH, STOUT
C
C          COMMON /GETCRC/  IEOF
C          LOGICAL          IEOF
C
C          COMMON /PACKEC/  NTZ, ISTAT, XLAT, XLONG,
1          IYR, INTINT, DDBT, IBEGH, NBS, IDOYSH(5), ISSHFT(5),
2          ISHFT, ALT, SKYCOV(12), ILOC
C
C          COMMON /PARAMS/  STOPIT, VERS, FIRST, NEWPAK, CALCGT
C          LOGICAL          STOPIT, VERS, FIRST, NEWPAK, CALCGT
C
C          COMMON /PCKINT/  KYR, KSTAT, CN(12), TG(12), KTIM, KARD(500), KYRL
C
C          COMMON /RAWDAT/  IDRY(24), IWET(24), IDEW(24), IPRESS(24),
1          IWNDSP(24), ICLAMT(24), ISOL(24), IDN(24),

```

```

2          IWNDIR(24),ICLTY(24),IRN(24),ISN(24),ICLTY1(24)
DIMENSION IRAW(24,12)
EQUIVALENCE      (IDRY(1),IRAW(1,1))
COMMON /LSTIME/  LSTHRS(24)
C
COMMON /REPORC/  IBEGM,IENDM
C
COMMON /UNDEF/   IUNDEF,UNDEF
C
C
DIMENSION DEABC(5)
C          DAY OF YEAR
IDOY = BEFORE(IMNTH) + IDAY
C          GET SUN PARAMETERS
CALL SUNPRM(IDOY,DEABC)
C          SOLAR CONSTANT
SOLCON = 435.2*(1.+0.033*COS(DTOR*360.*FLOAT(IDOY)/365.))
C          LOOP OVER HOURS IN THE DAY
DO 1000 IH=1,24
C          READ IN WEATHER DATA
READ (INWTH,9001) IVR,ID,IM,IHR,SKYCVR,IWINDR,WNDSPD,WSVECT,
1 IWTHR,DEP,PRESMB,TDRY,RELH,DIRH,IDIFFH,ILLUMH,IRH,IATMRH
9001 FORMAT(I1,3I2,1X,F4.2,1X,I3,2(1X,F4.1),1X,I2,1X,F4.1,1X,F6.1,
1 1X,F5.1,1X,F4.2,1X,I4,1X,I4,1X,I7,1X,I4,1X,I4)
C          CONVERT DRY-BULB FROM CENTIGRADE TO FAHRENHEIT
TDRYF = 1.8*TDRY + 32.
C          CALCULATE WET-BULB AND DEW POINT
C          SATURATED VAPOR PRESSURE
PS = PPWVMS(TDRYF)
C          PARTIAL PRESSURE
PW = RELH*PS
C          CONVERT PRESSURE FROM MILLIBARS TO INCHES OF HG
PRESHG = .02953*PRESMB
C          HUMIDITY RATIO
HUMRAT = 0.622*PW/(PRESHG-PW)
C          SPECIFIC ENTHALPY
ENTH = 0.24*TDRYF + (1061.+0.444*TDRYF)*HUMRAT
TWETF = WBF(ENTH,PRESHG)
Y = LOG(PW)
IF (PW .LE. 0.1836) THEN
    TDEWF = 71.98 + 24.873*Y + 0.8927*Y*Y
ELSE
    TDEWF = 79.047 + 30.579*Y + 1.8893*Y*Y
END IF
C          CONVERT WINDSPEED FROM M/S TO KNOTS
WSKNOT = 1.9438*WNDSPD
SOLHOR = 0.
SOLDRN = 0.
C          SET UPPER AND LOWER HOUR ANGLE BIN EDGES FOR SOLAR ZENITH
C          ANGLE CALCULATION. HOUR ANGLE IN UNITS OF HOURS.
UL = FLOAT(IH) - 12. + FLOAT(NTZ) + DEABC(2) - STALON/PIOV12
BL = UL -1.
C          SUNRISE AND SUNSET HOUR ANGLES
SSHA = ACOS(-TAN(STALAT)*TAN(DEABC(1)))/PIOV12
SRHA = -SSHA
C          SKIP IF SUN DOWN
IF ((UL .LE. SRHA) .OR. (BL .GE. SSHA)) GO TO 300
C          RESET BIN EDGES AT SUNRISE OR SUNSET
IF (SRHA .GT. BL) BL = SRHA
IF (SSHA .LT. UL) UL = SSHA
IF ((UL-BL) .LT. .02) GO TO 300
C          TOTAL HORIZONTAL SOLAR; CONVERT FROM W/M**2 TO
C          BTU/(FT**2)(HR)
SOLHOR = .31721*FLOAT(IDIRH+IDIFFH)
C          GET THE AVERAGE OF THE COSINE OF THE SOLAR ZENITH ANGLE

```

```

C          FOR THE 1 HOUR TIME BIN
          A = SIN(DEABC(1))*SIN(STALAT)
          B = COS(DEABC(1))*COS(STALAT)
          COSZIN = A*(UL-BL) + B*(SIN(PIOV12*UL)-SIN(PIOV12*BL))/PIOV12
          COSZAV = COSZIN/(UL-BL)
C          GET DIRECT NORMAL SOLAR
          SOLDRN = .31721*FLOAT(IDIRH)/COSZAV
C          PUT LIMITS ON THE SUNRISE AND SUNSET BEAM RADIATION
          CALL MAXDIR(COSZAV,SOLCON,DIRMAX)
          SOLDRN = AMIN1(SOLDRN,DIRMAX)
300 CONTINUE
C          FILL THE DATA ARRAYS
          IDRY(IH) = IROUND(TDRYF)
          IWET(IH) = MIN0(IDRY(IH),IROUND(TWETF))
          IDEW(IH) = MIN0(IWET(IH),IROUND(TDEWF))
          IPRESS(IH) = IROUND(100.*PRESHG)
          IWNDSP(IH) = IROUND(WSKNOT)
          IWNDIR(IH) = IROUND(.0444444*FLOAT(IWINDR))
          IF (IWNDIR(IH) .EQ. 16) IWNDIR(IH) = 0
          ICLAMT(IH) = IROUND(10.*SKYCVR)
          ISOL(IH) = IROUND(SOLHOR)
          IDN(IH) = IROUND(SOLDRN)
          ICLTY(IH) = 2
          ICLTY1(IH) = 2
          IRN(IH) = 0
          ISN(IH) = 0
1000 CONTINUE
          RETURN
          END

```

FORTRAN fragment: direct and diffuse solar calculation

```

C
C          Fortran fragment to calculate direct and diffuse
C          solar radiation from total solar and to get wet-bulb
C          and dew point temperatures from dry-bulb, relative humidity
C          and atmospheric pressure.
C
C          Start with: SOLHOR - total horiz. solar in Btu/hr-area
C                   TDRYF - dry-bulb temperature in degrees Fahrenheit
C                   PRESHG - atmospheric pressure in inches of mercury
C                   RELHUM - relative humidity in percent
C
C          Define:   DTOR - degrees to radians = pi/180
C                   PIOV12 - pi/12
C                   IH - hour of the day (1 - 24)
C                   IDOY - day of year (1 - 365)
C                   STALAT - weather station latitude in radians
C                   STALON - station longitude in radians
C                   XLONG - station longitude in degrees
C                   NTZ - time zone (PST=8, EST=5, Greenwich=0)
C
C          Externals: SUNPRM
C                   MAXDIR
C                   PPWVMS
C                   WBF
C
C                   These are all available from WTHPRC, the DOE-2
C                   weather processor.
C
C          DIMENSION DEABC(5)
C
C          DIRN = 0.
C          DIF = 0.
C
C          GET SOLAR CONSTANTS. DEABC(1) IS THE SOLAR
C          DECLINATION ANGLE; DEABC(2) IS THE EQUATION
C          OF TIME.
C          CALL SUNPRM(IDOY,DEABC)

```

```

C          GET SOLAR CONSTANT.  THIS FORMULA IS FROM KREIDER
C          AND RABL PAGE 237
C          BECKMAN, PAGE 7.
SOLCON = 435.2*(1. + 0.033*COS(DTOR*360.*FLOAT(IDOY)/365.))
C          GET HOUR ANGLE AT UPPER AND LOWER HOUR BIN EDGE
C          THIS IS LOCAL TIME!
UL      = FLOAT(IH) - 12. + FLOAT(NTZ) + DEABC(2) - XLONG/15.
BL      = UL - 1.
C          SUNRISE AND SUNSET HOUR ANGLES
SSHA    = ACOS(-TAN(STALAT)*TAN(DEABC(1)))/PIOV12
SRHA    = -SSHA
C          RESET BIN BOUNDARIES TO ALLOW FOR SUNRISE AND SET
IF ((UL .LE. SRHA) .OR. (BL .GE. SSHA)) GO TO 1100
IF (SRHA .GT. BL) BL = SRHA
IF (SSHA .LT. UL) UL = SSHA
IF ((UL-BL) .LT. 0.02) GO TO 1100
IF (SOLHOR .EQ. 0) GO TO 1100
A      = SIN(DEABC(1))*SIN(STALAT)
B      = COS(DEABC(1))*COS(STALAT)
C          INTEGRATE SOLAR Z DIREC. COSINE OVER BIN
COSZIN = A*(UL-BL) + B*(SIN(PIOV12*UL)-SIN(PIOV12*BL))/PIOV12
C          AVERAGE COSINE OF THE SOLAR ZENITH ANGLE FOR THE HOUR
COSZAV = COSZIN/(UL-BL)
C          EXTRATERRESTRIAL SOLAR HORIZONTAL
SOLEXH  = SOLCON*COSZIN
C          K sub T IS THE RATIO OF TERRESTRIAL TO EXTRATERRESTRIAL SOLAR
RKT = AMIN1(SOLHOR/SOLEXH,0.9)
C          GET DIFFUSE COMPONENT FROM ERBS, KLEIN, AND DUFFIE
C          CORRELATION
IF (RKT .LE. 0.22) DIF = SOLHOR*(1.-0.09*RKT)
RKT2   = RKT*RKT
IF ((RKT .GT. 0.22) .AND. (RKT .LE. 0.8)) DIF = SOLHOR*
1      (.9511-.1604*RKT+4.388*RKT2-16.638*RKT*RKT2+12.336*RKT2*RKT2)
IF (RKT .GT. 0.8) DIF = 0.165*SOLHOR
C          DIRECT HORIZONTAL
DIRH    = AMAX1(0.,SOLHOR-DIF)
C          DIRECT NORMAL
DIRN    = DIRH/COSZAV
C          CHECK FOR MAXIMUM DIRECT NORMAL
CALL MAXDIR(COSZAV,SOLCON,DIRMAX)
DIRN    = AMIN1(DIRN,DIRMAX)
1100 CONTINUE
C          CALCULATE WET-BULB AND DEW POINT.  THE PROCEDURE IS
C          BASICALLY THAT GIVEN ON PAGE 6.16, ASHRAE FUNDAMENTALS
C          1989, SITUATION NO. 3
C          SATURATED VAPOR PRESSURE
PS = PPWVMS(TDRYF)
C          PARTIAL PRESSURE
PW = .01*RELHUM*PS
C          HUMIDITY RATIO
HUMRAT = .622*PW/(PRESHG-PW)
C          SPECIFIC ENTHALPY
ENTH = .24*TDRYF + (1061.+ .444*TDRYF)*HUMRAT
C          WET-BULB TEMPERATURE
TWETF = WBF(ENTH,PRESHG)
C          DEW POINT TEMPERATURE
Y = LOG(PW)
IF (PW .LE. .1836) THEN
    TDEWF = 71.98 + 24.873*Y + .8927*Y*Y
ELSE
    TDEWF = 79.047 + 30.579*Y + 1.8893*Y*Y
END IF

```

Using Standard Formats for Measured Data

As an alternative to using the OTHER subroutine for measured or nonstandard data, one can convert the measured/nonstandard data into one of the standard formats and then run the DOE-2 weather processor on that format. For weather with no solar data, convert to the TRY format described below under "TRY Format," and then PACK with the weather processor using TRYSLM. Only the following TRY data fields need be filled (all unfilled fields should contain 9's):

hour	year	wet-bulb temperature	wind speed	type of lowest cloud layer
day	station number	dew point temperature	station pressure	present weather
month	dry-bulb temperature	wind direction	total sky cover	

For data sets with measured solar data, if global and direct are known, use the TMY2 format, described below under "TMY2 Format." The only hourly fields that need to be filled are:

hour	year	opaque sky cover	atmospheric pressure	present weather
day	global horizontal radiation	dry-bulb temperature	wind direction	
month	direct normal radiation	dew point temperature	wind speed	

In the TMY2 header record, all the data items should be filled. For data sets with global horizontal solar data and relative humidity instead of wet-bulb/dew point, use the DOE-2 measured data format described below.

Using the DOE-2 Measured Weather Format

Measured weather data can be put in the DOE-2 measured weather data format. This format uses metric units and contains the most commonly measured weather variables. For instance the format uses relative humidity rather than wet-bulb or dew-point temperature and global horizontal solar rather than direct & diffuse.

Format:

The file must be 8760 records in length, each record being 80 characters long.

Item	Columns	Format	Units
hour	1 - 2	I2	
day	3 - 4	I2	
month	5 - 6	I2	
year	7 -10	I4	
drybulb	11-20	F10.1	Centigrade
relative humidity	21-30	F10.2	a fraction, 0.0 to 1.0
atmos. pressure	31-40	F10.1	millibars
total horiz. rad.	41-50	F10.0	watts/m2
wind speed	51-60	F10.1	m/s
Wind dir.	61-70	F10.0	degrees from north (north=0;east=90)
sky cover	71-80	F10.1	a fraction 0.0 to 1.0

All quantities must be filled. If sky cover is unknown, put in -999.

The raw hourly weather file will look something like this (1 day only):

1	1	11997	20.0	0.6	1003.6	0.	4.5	273.	0.3
2	1	11997	21.0	0.6	1003.6	0.	4.5	273.	0.3
3	1	11997	22.0	0.6	1003.6	0.	4.5	273.	0.3
4	1	11997	23.0	0.6	1003.6	0.	4.5	273.	0.3
5	1	11997	24.0	0.6	1003.6	0.	4.5	273.	0.3
6	1	11997	25.0	0.6	1003.6	0.	4.5	273.	0.3
7	1	11997	26.0	0.6	1003.6	50.	4.5	273.	0.3
8	1	11997	27.0	0.6	1003.6	200.	4.5	273.	0.3
9	1	11997	28.0	0.6	1003.6	450.	4.5	273.	0.3
10	1	11997	29.0	0.6	1003.6	600.	4.5	273.	0.3
11	1	11997	29.0	0.6	1003.6	750.	4.5	273.	0.3
12	1	11997	29.0	0.6	1003.6	750.	4.5	273.	0.3
13	1	11997	28.0	0.6	1003.6	750.	4.5	273.	0.3
14	1	11997	27.0	0.6	1003.6	600.	4.5	273.	0.3
15	1	11997	26.0	0.6	1003.6	450.	4.5	273.	0.3
16	1	11997	25.0	0.6	1003.6	200.	4.5	273.	0.3

17	1	11997	24.0	0.6	1003.6	50.	4.5	273.	0.3
18	1	11997	23.0	0.6	1003.6	20.	4.5	273.	0.3
19	1	11997	23.0	0.6	1003.6	0.	4.5	273.	0.3
20	1	11997	21.0	0.6	1003.6	0.	4.5	273.	0.3
21	1	11997	19.0	0.6	1003.6	0.	4.5	273.	0.3
22	1	11997	18.0	0.6	1003.6	0.	4.5	273.	0.3
23	1	11997	17.0	0.6	1003.6	0.	4.5	273.	0.3
24	1	11997	16.0	0.6	1003.6	0.	4.5	273.	0.3

Example Input:

```
PACK
Test MeasWth Format
OTHER  -999  -999      5 40.7  74.2 30-BITSOLAR      4 20.  .025
1.00  1.00  1.00  1.00  1.00  1.00  1.00  1.00  1.00  1.00  1.00  1.00
-999.
LIST
PACKED  -999  -999      1  12
STAT
END
```

TRY Format

File Field Number	Columns	Element
001	01 - 05	STATION NUMBER
002	06 - 08	DRY-BULB TEMPERATURE
003	09 - 11	WET-BULB TEMPERATURE
004	12 - 14	DEW POINT TEMPERATURE
005	15 - 17	WIND DIRECTION
006	18 - 20	WIND SPEED
007	21 - 24	STATION PRESSURE
008	25	WEATHER
009	26 - 27	TOTAL SKY COVER
010	28 - 29	AMOUNT OF LOWEST CLOUD LAYER
011	30	TYPE OF LOWEST CLOUD OR OBSCURING PHENOMENA
012	31 - 33	HEIGHT OF BASE OF LOWEST LAYER
013	34 - 35	AMOUNT OF SECOND CLOUD LAYER
014	36	TYPE OF CLOUD - SECOND LAYER
015	37 - 39	HEIGHT OF BASE OF SECOND LAYER
016	40 - 41	SUMMATION AMOUNT OF FIRST TWO LAYERS
017	42 - 43	AMOUNT OF THIRD CLOUD LAYER
018	44	TYPE OF CLOUD - THIRD LAYER
019	45 - 47	HEIGHT OF BASE OF THIRD LAYER
020	48 - 49	SUMMATION AMOUNT OF FIRST THREE LAYERS
021	50 - 51	AMOUNT OF FOURTH CLOUD LAYER
022	52	TYPE OF CLOUD - FOURTH LAYER
023	53 - 55	HEIGHT OF BASE OF FOURTH LAYER
024	56 - 59	SOLAR RADIATION ^a
025	60 - 69	BLANK
026	70 - 73	YEAR
027	74 - 75	MONTH
028	76 - 77	DAY
029	78 - 79	HOUR
030	80	BLANK

^aThe DOE-2 weather processor recognizes the following solar data in TRY format:

Columns 57-59	Total horizontal radiation in Btu/ft ² -hr
Columns 61-63	Direct normal radiation in Btu/ft ² -hr

Packing a TRY file with solar data formatted as above and specifying the SOLAR option will result in the solar data being transferred to the DOE-2 packed weather file.

TRY Format (continued)

File Field Number	Columns	Element	Configuration	Code Definitions and Remarks
009	26 - 27	TOTAL SKY COVER	00 - 10	Amount of the sky covered by clouds or obscuring phenomena in tenths.
010	28 - 29	AMT OF LOWEST CLOUD LAYER	or 99	
013	34 - 35	AMT OF SECOND CLOUD LAYER		00-10 = 0-10 tenths
016	40 - 41	SUM OF FIRST TWO LAYERS		99 = Missing
017	42 - 43	AMT OF THIRD CLOUD LAYER		
020	48 - 49	SUM OF FIRST THREE LAYERS		
021	50 - 51	AMT OF FOURTH CLOUD LAYER		
011	30	TYPE OF LOWEST CLOUD OR OBSCURING PHENOMENA	0 - 9	Generic cloud type or obscuring phenomena: 0 = Clear
014	36	TYPE OF CLOUD - SECOND LAYER		1 = Fog or other obscuring phenomena 2 = Stratus or Fractus Stratus
018	44	TYPE OF CLOUD - THIRD LAYER		3 = Stratocumulus
022	52	TYPE OF CLOUD - FOURTH LAYER		4 = Cumulus or Cumulus Fractus 5 = Cumulonimbus or Mammatus 6 = Altostratus or Nimbostratus 7 = Altocumulus 8 = Cirrus 9 = Cirrostratus or Cirrocumulus 9 = Unknown if the amount of cloud is 99
012	31 - 33	HEIGHT OF BASE OF LOWEST LAYER	000 - 760	Height of base of clouds or obscuring phenomena in hundreds of feet
015	37 - 39	HEIGHT OF BASE OF SECOND LAYER	or 777	000-760 = 0-76,000 feet 777 = Unlimited - clear
019	45 - 47	HEIGHT OF BASE OF THIRD LAYER	or 888	888 = Cirroform clouds of unknown height 999 = Missing
023	53 - 55	HEIGHT OF BASE OF FOURTH LAYER	or 999	
024	56 - 59	SOLAR RADIATION	0000 - 1999 or 9999	Total solar radiation in Langleys to tenths. Values are for the hour ending at time indicated in Field 029. 0000-1999 = 0-199.9 Langleys 9999 = Missing
025	60 - 69	BLANK	△ △ △ △ △ △	<i>Blank field--reserved for future use.</i>
026	70 - 73	YEAR	1900 - 2099	Year
027	74 - 75	MONTH	01 - 12	Month of year: 01 = Jan., 02 = Feb., etc.
028	76 - 77	DAY	01 - 31	Day of month
029	78 - 79	HOUR	00 - 23	Hour of observation in Local Standard Time. 00-23 = 0000-2300 LST
030	80	BLANK	△ △ △ △ △ △	<i>Blank field, reserved for future use.</i>

CD144 Format

STATION NUMBER	YEAR	MO	DAY	HOUR	CEILING HEIGHT	CLOUD COVER BY LAYER				VISIBILITY	WEATHER/OBSTRUCTIONS								
						1st	2nd	3rd	4th		THUNDER-STORMS	RAIN	DRIZZLE	SNOW	SNOW SHOWERS	ICE PRECIP	OBSTR	OBSTR	
X X X X X	X X	X X	X X	X X	X X X	X	X	X	X	X X X	X	X	X	X	X	X	X	X	X
← 1 TO 31 →																			

SEA LEVEL PRESSURE	DEW POINT TEMPERATURE	WIND		STATIC PRESSURE	DRY-BULB TEMPERATURE	WET-BULB TEMPERATURE	RELATIVE HUMIDITY
		DIRECTION	SPEED				
X X X X	X X X	X X	X X	X X X X	X X X	X X X	X X X
← 32 TO 55 →							

CLOUDS AND OBSCURING PHENOMENA															FILLER	
TOTAL COVER	LAYER 1			LAYER 2			SUM AMOUNT	LAYER 3			SUM AMOUNT	LAYER 4				OPAQUE
	AMOUNT	TYPE	HEIGHT	AMOUNT	TYPE	HEIGHT		AMOUNT	TYPE	HEIGHT		AMOUNT	TYPE	HEIGHT		
X	X	X	X X X	X	X	X X X	X	X	X	X X X	X	X	X	X X X	X	
← 56 TO 80 →																

Figure 1: Record format for CD144 weather FILES

INDEX

- Weather Data in DOE-2.2
 - Example of subroutine OTHER, 11
 - Format
 - TMY2, 3
 - WYEC2, 4
 - Introduction, 2
 - Nonstandard Data, 10
 - Subroutine OTHER, 10
 - Solar Data, 3
 - Source Data, 2
 - Source Data and Formats (TMY2 and WYEC2), 2
 - Standard Formats, Measured Data, 15
- TRY Format Example, 17
- Using the DOE-2 Measured Weather Format, 15
- Variables (Hourly, Hourly Solar, Variable, Header Data), 2
- Weather Processor, 4
 - ASCII Weather Files
 - FMTWTH Program Listing, 7
 - Moving Data, 7
 - WTHFMT Program Listing, 9
 - I/O, 4
 - Input Description, 4

